

# NAG Fortran Library Routine Document

## F08SBF (DSYGVX)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

F08SBF (DSYGVX) computes selected eigenvalues, and optionally, eigenvectors of a real generalized symmetric-definite eigenproblem, of the form

$$Az = \lambda Bz, \quad ABz = \lambda z \quad \text{or} \quad BAz = \lambda z,$$

where  $A$  and  $B$  are symmetric and  $B$  is also positive-definite. Eigenvalues and eigenvectors can be selected by specifying either a range of values or a range of indices for the desired eigenvalues.

### 2 Specification

```

SUBROUTINE F08SBF ( ITYPE, JOBZ, RANGE, UPLO, N, A, LDA, B, LDB, VL, VU,
1 IL, IU, ABSTOL, M, W, Z, LDZ, WORK, LWORK, IWORK,
2 JFAIL, INFO)
    INTEGER          ITYPE, N, LDA, LDB, IL, IU, M, LDZ, LWORK, IWORK(*),
1 JFAIL(*), INFO
    double precision A(LDA,*), B(LDA,*), VL, VU, ABSTOL, W(*), Z(LDZ,*),
1 WORK(*)
    CHARACTER*1     JOBZ, RANGE, UPLO

```

The routine may be called by its LAPACK name *dsygvx*.

### 3 Description

F08SBF (DSYGVX) first performs a Cholesky factorization of the matrix  $B$  as  $B = U^T U$ , when UPLO = 'U' or  $B = LL^T$ , when UPLO = 'L'. The generalized problem is then reduced to a standard symmetric eigenvalue problem

$$Cx = \lambda x,$$

which is solved for the desired eigenvalues and eigenvectors. The eigenvectors of  $C$  are then backtransformed to give the eigenvectors of the original problem.

For the problem  $Az = \lambda Bz$  and  $ABz = \lambda z$ , the eigenvectors are normalized so that

$$z^T Bz = I.$$

For the problem  $BAz = \lambda z$  we correspondingly have

$$z^T B^{-1} z = I.$$

### 4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia URL: <http://www.netlib.org/lapack/lug>

Demmel J W and Kahan W (1990) Accurate singular values of bidiagonal matrices *SIAM J. Sci. Statist. Comput.* **11** 873–912

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

## 5 Parameters

- 1: ITYPE – INTEGER *Input*  
*On entry:* specifies the problem type to be solved:  
     if ITYPE = 1,  $Az = \lambda Bz$ ;  
     if ITYPE = 2,  $ABz = \lambda z$ ;  
     if ITYPE = 3,  $BAz = \lambda z$ .
- 2: JOBZ – CHARACTER\*1 *Input*  
*On entry:* if JOBZ = 'N', compute eigenvalues only.  
 If JOBZ = 'V', compute eigenvalues and eigenvectors.  
*Constraint:* JOBZ = 'N' or 'V'.
- 3: RANGE – CHARACTER\*1 *Input*  
*On entry:* if RANGE = 'A', all eigenvalues will be found.  
 If RANGE = 'V', all eigenvalues in the half-open interval (VL, VU] will be found.  
 If RANGE = 'I', the ILth to IUth eigenvalues will be found.
- 4: UPLO – CHARACTER\*1 *Input*  
*On entry:* if UPLO = 'U', the upper triangles of  $A$  and  $B$  are stored.  
 If UPLO = 'L', the lower triangles of  $A$  and  $B$  are stored.
- 5: N – INTEGER *Input*  
*On entry:*  $n$ , the order of the matrix pencil ( $A, B$ ).  
*Constraint:*  $N \geq 0$ .
- 6: A(LDA,\*) – **double precision** array *Input/Output*  
**Note:** the second dimension of the array  $A$  must be at least  $\max(1, N)$ .  
*On entry:* the  $n$  by  $n$  symmetric matrix  $A$ .  
     If UPLO = 'U', the leading  $n$  by  $n$  upper triangular part of  $A$  contains the upper triangular part of the matrix  $A$ .  
     If UPLO = 'L', the leading  $n$  by  $n$  lower triangular part of  $A$  contains the lower triangular part of the matrix  $A$ .  
*On exit:* the lower triangle (if UPLO = 'L') or the upper triangle (if UPLO = 'U') of  $A$ , including the diagonal, is destroyed.
- 7: LDA – INTEGER *Input*  
*On entry:* the first dimension of the array  $A$  as declared in the (sub)program from which F08SBF (DSYGVX) is called.  
*Constraint:*  $LDA \geq \max(1, N)$ .
- 8: B(LDA,\*) – **double precision** array *Input/Output*  
**Note:** the second dimension of the array  $B$  must be at least  $\max(1, N)$ .  
*On entry:* the symmetric matrix  $B$ :  
     if UPLO = 'U', the leading  $n$  by  $n$  upper triangular part of  $B$  contains the upper triangular part of the matrix  $B$ ;  
     if UPLO = 'L', the leading  $n$  by  $n$  lower triangular part of  $B$  contains the lower triangular part of the matrix  $B$ .

*On exit:* if  $\text{INFO} \leq N$ , the part of  $B$  containing the matrix is overwritten by the triangular factor  $U$  or  $L$  from the Cholesky factorization  $B = U^T U$  or  $B = LL^T$ .

9: LDB – INTEGER *Input*

*On entry:* the first dimension of the array  $B$  as declared in the (sub)program from which F08SBF (DSYGVX) is called.

*Constraint:*  $\text{LDB} \geq \max(1, N)$ .

10: VL – *double precision* *Input*

11: VU – *double precision* *Input*

*On entry:* if  $\text{RANGE} = 'V'$ , the lower and upper bounds of the interval to be searched for eigenvalues.

*Constraint:*  $\text{VL} < \text{VU}$ .

If  $\text{RANGE} = 'A'$  or  $'I'$ , VL and VU are not referenced.

12: IL – INTEGER *Input*

13: IU – INTEGER *Input*

*On entry:* if  $\text{RANGE} = 'I'$ , the indices (in ascending order) of the smallest and largest eigenvalues to be returned.

If  $\text{RANGE} = 'A'$  or  $'V'$ , IL and IU are not referenced.

*Constraints:*

if  $N = 0$ ,  $\text{IL} = 1$  and  $\text{IU} = 0$ ;

if  $N > 0$ ,  $1 \leq \text{IL} \leq \text{IU} \leq N$ .

14: ABSTOL – *double precision* *Input*

*On entry:* the absolute error tolerance for the eigenvalues. An approximate eigenvalue is accepted as converged when it is determined to lie in an interval  $[a, b]$  of width less than or equal to

$$\text{ABSTOL} + \epsilon \max(|a|, |b|),$$

where  $\epsilon$  is the *machine precision*. If ABSTOL is less than or equal to zero, then  $\epsilon \|T\|_1$  will be used in its place, where  $T$  is the tridiagonal matrix obtained by reducing  $C$  to tridiagonal form. Eigenvalues will be computed most accurately when ABSTOL is set to twice the underflow threshold  $2 \times \text{X02AMF}()$ , not zero. If this routine returns with  $\text{INFO} > 0$ , indicating that some eigenvectors did not converge, try setting ABSTOL to  $2 \times \text{X02AMF}()$ . See Demmel and Kahan (1990).

15: M – INTEGER *Output*

*On exit:* the total number of eigenvalues found.

If  $\text{RANGE} = 'A'$ ,  $M = N$ .

If  $\text{RANGE} = 'I'$ ,  $M = \text{IU} - \text{IL} + 1$ .

*Constraint:*  $0 \leq M \leq N$ .

16: W(\*) – *double precision* array *Output*

**Note:** the dimension of the array  $W$  must be at least  $\max(1, N)$ .

*On exit:* the first  $M$  elements contain the selected eigenvalues in ascending order.

- 17:  $Z(LDZ,*)$  – *double precision* array *Output*  
**Note:** the second dimension of the array  $Z$  must be at least  $\max(1, M)$ .  
*On exit:* if  $JOBZ = 'V'$ , then if  $INFO = 0$ , the first  $m$  columns of  $Z$  contain the orthonormal eigenvectors of the matrix  $A$  corresponding to the selected eigenvalues, with the  $i$ th column of  $Z$  holding the eigenvector associated with  $W(i)$ . The eigenvectors are normalised as follows:  
     if  $ITYPE = 1$  or  $2$ ,  $Z^T B Z = I$ ;  
     if  $ITYPE = 3$ ,  $Z^T B^{-1} Z = I$ .  
 If  $JOBZ = 'N'$ ,  $Z$  is not referenced.  
 If an eigenvector fails to converge, then that column of  $Z$  contains the latest approximation to the eigenvector, and the index of the eigenvector is returned in  $JFAIL$ .  
**Note:** the user must ensure that at least  $\max(1, M)$  columns are supplied in the array  $Z$ ; if  $RANGE = 'V'$ , the exact value of  $M$  is not known in advance and an upper bound must be used.
- 18:  $LDZ$  – INTEGER *Input*  
*On entry:* the first dimension of the array  $Z$  as declared in the (sub)program from which F08SBF (DSYGVX) is called.  
*Constraints:*  
     if  $JOBZ = 'V'$ ,  $LDZ \geq \max(1, N)$ ;  
      $LDZ \geq 1$  otherwise.
- 19:  $WORK(*)$  – *double precision* array *Workspace*  
**Note:** the dimension of the array  $WORK$  must be at least  $\max(1, LWORK)$ .  
*On exit:* if  $INFO = 0$ ,  $WORK(1)$  returns the optimal  $LWORK$ .
- 20:  $LWORK$  – INTEGER *Input*  
*On entry:* the dimension of the array  $WORK$  as declared in the (sub)program from which F08SBF (DSYGVX) is called.  
 For optimal efficiency,  $LWORK \geq (nb + 3) \times N$ , where  $nb$  is the optimal block size for F08FEF (DSYTRD).  
 If  $LWORK = -1$ , a workspace query is assumed; the routine only calculates the optimal size of the  $WORK$  array, returns this value as the first entry of the  $WORK$  array, and no error message related to  $LWORK$  is issued.  
*Constraint:*  $LWORK \geq \max(1, 8 \times N)$ .
- 21:  $IWORK(*)$  – INTEGER array *Workspace*  
**Note:** the dimension of the array  $IWORK$  must be at least  $\max(1, 5 \times N)$ .
- 22:  $JFAIL(*)$  – INTEGER array *Output*  
**Note:** the dimension of the array  $JFAIL$  must be at least  $\max(1, N)$ .  
*On exit:* if  $JOBZ = 'V'$ , then if  $INFO = 0$ , the first  $M$  elements of  $JFAIL$  are zero. If  $INFO > 0$ ,  $JFAIL$  contains the indices of the eigenvectors that failed to converge.  
 If  $JOBZ = 'N'$ ,  $JFAIL$  is not referenced.
- 23:  $INFO$  – INTEGER *Output*  
*On exit:*  $INFO = 0$  unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO =  $-i$ , the  $i$ th argument had an illegal value.

INFO > 0

F07FDF (DPOTRF) or F08FBB (DSYEVS) returned an error code:

$\leq N$  if INFO =  $i$ , F08FBB (DSYEVS) failed to converge;  $i$  eigenvectors failed to converge. Their indices are stored in array JFAIL;

$> N$  if INFO =  $N + i$ , for  $1 \leq i \leq N$ , then the leading minor of order  $i$  of  $B$  is not positive-definite. The factorization of  $B$  could not be completed and no eigenvalues or eigenvectors were computed.

## 7 Accuracy

If  $B$  is ill-conditioned with respect to inversion, then the error bounds for the computed eigenvalues and vectors may be large, although when the diagonal elements of  $B$  differ widely in magnitude the eigenvalues and eigenvectors may be less sensitive than the condition of  $B$  would suggest. See Section 4.10 of Anderson *et al.* (1999) for details of the error bounds.

## 8 Further Comments

The total number of floating-point operations is proportional to  $n^3$ .

The complex analogue of this routine is F08SPF (ZHEGVX).

## 9 Example

To find the eigenvalues in the half-open interval  $(-1.0, 1.0]$ , and corresponding eigenvectors, of the generalized symmetric eigenproblem  $Az = \lambda Bz$ , where

$$A = \begin{pmatrix} 0.24 & 0.39 & 0.42 & -0.16 \\ 0.39 & -0.11 & 0.79 & 0.63 \\ 0.42 & 0.79 & -0.25 & 0.48 \\ -0.16 & 0.63 & 0.48 & -0.03 \end{pmatrix}$$

and

$$B = \begin{pmatrix} 4.16 & -3.12 & 0.56 & -0.10 \\ -3.12 & 5.03 & -0.83 & 1.09 \\ 0.56 & -0.83 & 0.76 & 0.34 \\ -0.10 & 1.09 & 0.34 & 1.18 \end{pmatrix}.$$

The example program for F08SCF (DSYGV) illustrates solving a generalized symmetric eigenproblem of the form  $ABz = \lambda z$ .

## 9.1 Program Text

**Note:** the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```

*      F08SBF Example Program Text
*      Mark 21. NAG Copyright 2004.
*      .. Parameters ..
INTEGER          NIN, NOUT
PARAMETER        (NIN=5,NOUT=6)
INTEGER          NB, NMAX, MMAX
PARAMETER        (NB=64,NMAX=10,MMAX=5)
INTEGER          LDA, LDB, LDZ, LWORK
PARAMETER        (LDA=NMAX,LDB=NMAX,LDZ=NMAX,LWORK=(NB+3)*NMAX)
DOUBLE PRECISION ZERO
PARAMETER        (ZERO=0.0D+0)
*      .. Local Scalars ..
DOUBLE PRECISION ABSTOL, VL, VU
INTEGER          I, IFAIL, IL, INFO, IU, J, LWKOPT, M, N
*      .. Local Arrays ..
DOUBLE PRECISION A(LDA,NMAX), B(LDB,NMAX), W(NMAX), WORK(LWORK),
+              Z(LDZ,MMAX)
INTEGER          INDEX(NMAX), IWORK(5*NMAX)
*      .. External Subroutines ..
EXTERNAL         DSYGVX, X04CAF
*      .. Executable Statements ..
WRITE (NOUT,*) 'F08SBF Example Program Results'
WRITE (NOUT,*)
*      Skip heading in data file
READ (NIN,*)
READ (NIN,*) N
IF (N.LE.NMAX) THEN
*
*      Read the lower and upper bounds of the interval to be searched,
*      and read the upper triangular parts of the matrices A and B
*
      READ (NIN,*) VL, VU
      READ (NIN,*) ((A(I,J),J=I,N),I=1,N)
      READ (NIN,*) ((B(I,J),J=I,N),I=1,N)
*
*      Set the absolute error tolerance for eigenvalues. With ABSTOL
*      set to zero, the default value is used instead
*
      ABSTOL = ZERO
*
*      Solve the generalized symmetric eigenvalue problem
*      A*x = lambda*B*x (ITYPE = 1)
*
      CALL DSYGVX(1,'Vectors','Values in range','Upper',N,A,LDA,B,
+              LDB,VL,VU,IL,IU,ABSTOL,M,W,Z,LDZ,WORK,LWORK,IWORK,
+              INDEX,INFO)
      LWKOPT = WORK(1)
*
      IF (INFO.GE.0 .AND. INFO.LE.N .AND. M.LE.MMAX) THEN
*
*      Print solution
*
      WRITE (NOUT,99999) 'Number of eigenvalues found =', M
      WRITE (NOUT,*)
      WRITE (NOUT,*) 'Eigenvalues'
      WRITE (NOUT,99998) (W(J),J=1,M)
*
      IFAIL = 0
      CALL X04CAF('General',' ',N,M,Z,LDZ,'Selected eigenvectors',
+              IFAIL)
      IF (INFO.GT.0) THEN
        WRITE (NOUT,99999)
        'INFO eigenvectors failed to converge, INFO =', INFO
        WRITE (NOUT,*)
        'Indices of eigenvectors that did not converge'
      END IF
    END IF
  
```

```

        WRITE (NOUT,99997) (INDEX(J),J=1,M)
      END IF
    ELSE IF (INFO.GT.N .AND. INFO.LE.2*N) THEN
      I = INFO - N
      WRITE (NOUT,99996) 'The leading minor of order ', I,
+      ' of B is not positive definite'
    ELSE IF (M.GT.MMAX) THEN
      WRITE (NOUT,99995) 'M greater than MMAX, M =', M,
+      ', MMAX =', MMAX
    ELSE
      WRITE (NOUT,99999) 'Failure in DSYGVX. INFO =', INFO
    END IF
*
*   Print workspace information
*
    IF (LWORK.LT.LWKOPT) THEN
      WRITE (NOUT,*)
      WRITE (NOUT,99994) 'Optimum workspace required = ', LWKOPT,
+      'Workspace provided          = ', LWORK
    END IF
  ELSE
    WRITE (NOUT,*)
    WRITE (NOUT,*) 'NMAX too small'
  END IF
  STOP
*
99999 FORMAT (1X,A,I5)
99998 FORMAT (3X,(8F8.4))
99997 FORMAT (3X,(8I8))
99996 FORMAT (1X,A,I4,A)
99995 FORMAT (1X,A,I5,A,I5)
99994 FORMAT (1X,A,I5,/1X,A,I5)
  END

```

## 9.2 Program Data

F08SBF Example Program Data

```

  4                               :Value of N

-1.0    1.0                       :Values of VL and VU

  0.24   0.39   0.42  -0.16
    -0.11  0.79   0.63
          -0.25   0.48
          -0.03 :End of matrix A

  4.16  -3.12   0.56  -0.10
        5.03  -0.83   1.09
          0.76   0.34
          1.18 :End of matrix B

```

## 9.3 Program Results

F08SBF Example Program Results

Number of eigenvalues found = 2

Eigenvalues

-0.4548 0.1001

Selected eigenvectors

```

      1      2
  1  0.3080  0.4469
  2  0.5329  0.0371
  3 -0.3496 -0.0505
  4 -0.6211 -0.4743

```